

# Chapter 5

## Relational algebraic ornamentation

This chapter turns to the **synthetic** direction of the interconnection between internalism and externalism. As stated in ??, internalist types can be hard to read and write, and it would be helpful to be able to switch to an alternative language for understanding and deriving internalist types. The alternative language adopted in this chapter is the **relational** language (Section 5.1), of which Bird and de Moor [1997] gave an authoritative account. Unlike the datatype declaration language, using relations we can give concise yet computationally intuitive specifications, which are amenable to manipulation by algebraic laws and theorems. A particularly expressive relational construction is the **relational fold**, and when fixing a basic datatype and casting the relational fold as the externalist predicate, we can synthesise a corresponding internalist datatype on the other side of the conversion isomorphism. More specifically, every relational algebra gives rise to an **algebraic ornamentation** (Section 5.2), whose optimised predicate (??) can be swapped (??) for the relational fold with the algebra. Specifications involving relational folds can then be met by constructing internalist programs whose types involve corresponding algebraic ornamented datatypes. Several examples are given in Section 5.3, followed by some discussion in Section 5.4.

## 5.1 Relational programming in Agda

Functional programs are known for their amenability to algebraic calculation (see, e.g., Backus [1978] and Bird [2010]), leading to one form of program correctness by construction: one begins with a specification in the form of a functional program that expresses a straightforward but possibly inefficient computation, and transforms it into an extensionally equal but more efficient functional program by applying algebraic laws and theorems. Using functional programs as the specification language means that specifications are directly executable, but the deterministic nature of functional programs can result in less flexible specifications. For example, when specifying an optimisation problem using a functional program that generates all feasible solutions and chooses an optimal one among them, the program necessarily enforces a particular way of choosing the optimal solution, but such enforcement should not in general be part of the specification. To gain more flexibility, the specification language was later generalised to **relational programs** (see, e.g., Bird [1996]). With relational programs, we specify only the relationship between input and output without actually specifying a way to execute the programs, so specifications in the form of relational programs can be as flexible as possible. Though lacking a directly executable semantics, most relational programs can still be read computationally as potentially partial and nondeterministic mappings, so relational specifications largely remain computationally intuitive as functional specifications.

To emphasise the computational interpretation of relations, we will mainly model a relation between sets  $A$  and  $B$  as a function sending each inhabitant of  $A$  to a subset of  $B$ . We define subsets by

$$\begin{aligned}\mathcal{P} &: \text{Set} \rightarrow \text{Set}_1 \\ \mathcal{P}A &= A \rightarrow \text{Set}\end{aligned}$$

That is, a subset  $s : \mathcal{P}A$  is a characteristic function that assigns a type to each inhabitant of  $A$ , and  $a : A$  is considered to be a member of  $s$  if the type  $s\ a : \text{Set}$  is inhabited. We may regard  $\mathcal{P}A$  as the type of computations that nondeterministically produce an inhabitant of  $A$ . A simple example is

$$\begin{aligned}\text{any} &: \{A : \text{Set}\} \rightarrow \mathcal{P}A \\ \text{any} &= \text{const } \top\end{aligned}$$

The subset  $any : \mathcal{P}A$  associates the unit type  $\top$  with every inhabitant of  $A$ . Since  $\top$  is inhabited,  $any$  can produce any inhabitant of  $A$ . While  $\mathcal{P}$  cannot be made into a conventional monad [Moggi, 1991; Wadler, 1992] because it is not an endofunctor, it can still be equipped with the usual monadic programming combinators (giving rise to a “relative monad” [Altenkirch et al., 2010]):

- The monadic unit is defined as

$$\begin{aligned} return &: \{A : \text{Set}\} \rightarrow A \rightarrow \mathcal{P}A \\ return &= \_ \equiv \_ \end{aligned}$$

The subset  $return\ a : \mathcal{P}A$  for some  $a : A$  simplifies to  $\lambda a' \mapsto a \equiv a'$ , so  $a$  is the only member of the subset.

- The monadic bind is defined as

$$\begin{aligned} \_ \gg= \_ &: \{A\ B : \text{Set}\} \rightarrow \mathcal{P}A \rightarrow (A \rightarrow \mathcal{P}B) \rightarrow \mathcal{P}B \\ \_ \gg= \_ \{A\} sf &= \lambda b \mapsto \Sigma[a : A] \ s\ a \times f\ a\ b \end{aligned}$$

If  $s : \mathcal{P}A$  and  $f : A \rightarrow \mathcal{P}B$ , then the subset  $s \gg= f : \mathcal{P}B$  is the disjoint union of all the subsets  $f\ a : \mathcal{P}B$  where  $a$  ranges over the inhabitants of  $A$  that belong to  $s$ ; that is, an inhabitant  $b : B$  is a member of  $s \gg= f$  exactly when there exists some  $a : A$  belonging to  $s$  such that  $b$  is a member of  $f\ a$ .

(We omit the proofs that the two combinators satisfy the (relative) monad laws up to pointwise isomorphism.) On top of  $return$  and  $\_ \gg= \_$ , the functorial map on  $\mathcal{P}$  is defined as

$$\begin{aligned} \_ \langle \$ \rangle &: \{A\ B : \text{Set}\} \rightarrow (A \rightarrow B) \rightarrow \mathcal{P}A \rightarrow \mathcal{P}B \\ f \langle \$ \rangle s &= s \gg= \lambda a \mapsto return\ (f\ a) \end{aligned}$$

and we also define a two-argument version for convenience:

$$\begin{aligned} \_ \langle \$ \rangle^2 &: \{A\ B\ C : \text{Set}\} \rightarrow (A \rightarrow B \rightarrow C) \rightarrow \mathcal{P}A \rightarrow \mathcal{P}B \rightarrow \mathcal{P}C \\ f \langle \$ \rangle^2 s\ t &= s \gg= \lambda a \mapsto t \gg= \lambda b \mapsto return\ (f\ a\ b) \end{aligned}$$

(The notation is a reference to applicative functors [McBride and Paterson, 2008], allowing us to think of functorial maps of  $\mathcal{P}$  as applications of pure functions to effectful arguments.)

We define a relation between two families of sets as a family of relations between corresponding sets in the families:

$$\begin{aligned} \_ \rightsquigarrow \_ &: \{I : \text{Set}\} \rightarrow (I \rightarrow \text{Set}) \rightarrow (I \rightarrow \text{Set}) \rightarrow \text{Set}_1 \\ \_ \rightsquigarrow \_ \{I\} X Y &= \{i : I\} \rightarrow X i \rightarrow \mathcal{P}(Y i) \end{aligned}$$

which is the usual generalisation of  $\_ \Rightarrow \_$  to allow nondeterminacy. Below we define several relational operators that we will need.

- Since functions are deterministic relations, we have the following combinator *fun* that lifts functions to relations using *return*.

$$\begin{aligned} \text{fun} &: \{I : \text{Set}\} \{X Y : I \rightarrow \text{Set}\} \rightarrow (X \Rightarrow Y) \rightarrow (X \rightsquigarrow Y) \\ \text{fun } f \ x &= \text{return } (f \ x) \end{aligned}$$

- The identity relation is just the identity function lifted by *fun*.

$$\begin{aligned} \text{idR} &: \{I : \text{Set}\} \{X : I \rightarrow \text{Set}\} \rightarrow (X \rightsquigarrow X) \\ \text{idR} &= \text{fun id} \end{aligned}$$

- Composition of relations is easily defined with  $\_ \gg \_$ : computing  $R \cdot S$  on input  $x$  is first computing  $S \ x$  and then feeding the result to  $R$ .

$$\begin{aligned} \_ \bullet \_ &: \{I : \text{Set}\} \{X Y Z : I \rightarrow \text{Set}\} \rightarrow (Y \rightsquigarrow Z) \rightarrow (X \rightsquigarrow Y) \rightarrow (X \rightsquigarrow Z) \\ (R \bullet S) \ x &= S \ x \gg R \end{aligned}$$

- Some relations do not carry obvious computational meaning, but can still be defined pointwise, like the **meet** of two relations:

$$\begin{aligned} \_ \cap \_ &: \{I : \text{Set}\} \{X Y : I \rightarrow \text{Set}\} \rightarrow (X \rightsquigarrow Y) \rightarrow (X \rightsquigarrow Y) \rightarrow (X \rightsquigarrow Y) \\ (R \cap S) \ x \ y &= R \ x \ y \times S \ x \ y \end{aligned}$$

- Unlike a function, which distinguishes between input and output, inherently a relation treats its domain and codomain symmetrically. This is reflected by the presence of the following **converse** operator:

$$\begin{aligned} \_ \circ &: \{I : \text{Set}\} \{X Y : I \rightarrow \text{Set}\} \rightarrow (X \rightsquigarrow Y) \rightarrow (Y \rightsquigarrow X) \\ (R \circ) \ y \ x &= R \ x \ y \end{aligned}$$

A relation can thus be “run backwards” simply by taking its converse. The nondeterministic and bidirectional nature of relations makes them a powerful and concise language for specifications, as will be demonstrated in Sections 5.3.2 and 5.3.3.

- We will also need **relators**, i.e., functorial maps on relations:

$$\begin{aligned}
\text{mapR} &: \{I : \text{Set}\} (D : \text{RDesc } I) \{X \ Y : I \rightarrow \text{Set}\} \rightarrow \\
&\quad (X \rightsquigarrow Y) \rightarrow \llbracket D \rrbracket X \rightarrow \mathcal{P}(\llbracket D \rrbracket Y) \\
\text{mapR } (\text{v } []) &\quad R \blacksquare = \text{return } \blacksquare \\
\text{mapR } (\text{v } (i :: is)) &R (x, xs) = \_,\_ \langle \$ \rangle^2 (R x) (\text{mapR } (\text{v } is) R xs) \\
\text{mapR } (\sigma S D) &R (s, xs) = (\_,\_ s) \langle \$ \rangle (\text{mapR } (D s) R xs) \\
\mathbb{R} &: \{I : \text{Set}\} (D : \text{Desc } I) \{X \ Y : I \rightarrow \text{Set}\} \rightarrow (X \rightsquigarrow Y) \rightarrow (\mathbb{F} D X \rightsquigarrow \mathbb{F} D Y) \\
\mathbb{R} D R \{i\} &= \text{mapR } (D i) R
\end{aligned}$$

Figure 5.1 Definition for relators.

$$\begin{aligned}
\mathbb{R} &: \{I : \text{Set}\} (D : \text{Desc } I) \{X \ Y : I \rightarrow \text{Set}\} \rightarrow \\
&\quad (X \rightsquigarrow Y) \rightarrow (\mathbb{F} D X \rightsquigarrow \mathbb{F} D Y)
\end{aligned}$$

If  $R : X \rightsquigarrow Y$ , the relation  $\mathbb{R} D R : \mathbb{F} D X \rightsquigarrow \mathbb{F} D Y$  applies  $R$  to the recursive positions of its input, leaving everything else intact. The definition of  $\mathbb{R}$  is shown in Figure 5.1. For example, if  $D = \text{ListD } A$ , then  $\mathbb{R} (\text{ListD } A)$  is, up to isomorphism,

$$\begin{aligned}
\mathbb{R} (\text{ListD } A) &: \{X \ Y : I \rightarrow \text{Set}\} \rightarrow \\
&\quad (X \rightsquigarrow Y) \rightarrow (\mathbb{F} (\text{ListD } A) X \rightsquigarrow \mathbb{F} (\text{ListD } A) Y) \\
\mathbb{R} (\text{ListD } A) R ('nil, \blacksquare) &= \text{return } ('nil, \blacksquare) \\
\mathbb{R} (\text{ListD } A) R ('cons, a, x, \blacksquare) &= (\lambda y \mapsto 'cons, a, y, \blacksquare) \langle \$ \rangle (R x)
\end{aligned}$$

Laws and theorems about relational programs are formulated with relational inclusion:

$$\begin{aligned}
\_\subseteq\_ &: \{I : \text{Set}\} \{X \ Y : I \rightarrow \text{Set}\} (X \rightsquigarrow Y) \rightarrow (X \rightsquigarrow Y) \rightarrow \text{Set} \\
\_\subseteq\_ \{I\} R S &= \{i : I\} \rightarrow (x : X i) (y : Y i) \rightarrow R x y \rightarrow S x y
\end{aligned}$$

or equivalence of relations, i.e., two-way inclusion:

$$\begin{aligned}
\_\simeq\_ &: \{I : \text{Set}\} \{X \ Y : I \rightarrow \text{Set}\} (R S : X \rightsquigarrow Y) \rightarrow \text{Set} \\
R \simeq S &= (R \subseteq S) \times (R \supseteq S)
\end{aligned}$$

where  $R \supseteq S$  is defined to be  $S \subseteq R$  as usual. For example, a relator preserves identity and composition, i.e.,

$$\mathbb{R} D \text{idR} \simeq \text{idR} \quad \text{and} \quad \mathbb{R} D (R \cdot S) \simeq \mathbb{R} D R \cdot \mathbb{R} D S$$

and is monotonic, i.e.,

$$\mathbb{R} D R \subseteq \mathbb{R} D S \quad \text{whenever} \quad R \subseteq S$$

Also, many concepts can be expressed in a surprisingly concise way with relational inclusion. For example, a relation  $R$  is a preorder if it is reflexive and transitive. In relational terms, these two conditions are expressed simply as

$$idR \subseteq R \quad \text{and} \quad R \cdot R \subseteq R$$

and are easily manipulable in calculations. Another important notion is **monotonic algebras** [Bird and de Moor, 1997, Section 7.2]: an algebra  $S : \mathbb{F} D X \rightsquigarrow X$  is **monotonic** on  $R : X \rightsquigarrow X$  (usually an ordering) if

$$S \cdot \mathbb{R} D R \subseteq R \cdot S$$

which says that if two input values to  $S$  have their recursive positions related by  $R$  and are otherwise equal, then the output values would still be related by  $R$ . (For example, let

$$D = \lambda \{ \blacksquare \mapsto v (\blacksquare :: \blacksquare :: []) \} : \text{Desc } \top$$

be a trivially indexed description with two recursive positions, and define

$$plus = fun (\lambda \{ (x, y, \blacksquare) \mapsto x + y \}) : \mathbb{F} D (const \text{ Nat}) \rightsquigarrow const \text{ Nat}$$

Then *plus* is monotonic on

$$leq = \lambda x y \mapsto y \leq x : const \text{ Nat} \rightsquigarrow const \text{ Nat}$$

which maps a natural number  $x$  to any natural number  $y$  that is at most  $x$ . Pointwise, the monotonicity statement expands to

$$(x \ y \ x' \ y' : \text{Nat}) \rightarrow (x \leq x') \times (y \leq y') \rightarrow x + y \leq x' + y'$$

i.e., addition is monotonic on its two arguments.) In the context of optimisation problems, monotonicity can be used to capture the **principle of optimality**, as will be shown in Section 5.3.3. Section 5.3.1 contains some simple relational calculations involving the above properties.

Having defined relations as nondeterministic mappings, it is straightforward to rewrite the datatype-generic *fold* with the subset combinators to obtain a relational version, which is denoted by the “banana bracket” [Meijer et al., 1991]:

**mutual**

$$\begin{aligned}
\llbracket - \rrbracket &: \{I : \text{Set}\} \{D : \text{Desc } I\} \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \rightsquigarrow X) \rightarrow (\mu D \rightsquigarrow X) \\
\llbracket - \rrbracket \{I\} \{D\} R \{i\} (\text{con } ds) &= \text{mapFoldR } D (D i) R ds \gg R \\
\text{mapFoldR} &: \{I : \text{Set}\} (D : \text{Desc } I) (D' : \text{RDesc } I) \rightarrow \\
&\quad \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \rightsquigarrow X) \rightarrow \llbracket D' \rrbracket (\mu D) \rightarrow \mathcal{P}(\llbracket D' \rrbracket X) \\
\text{mapFoldR } D (\vee []) \quad R \blacksquare &= \text{return } \blacksquare \\
\text{mapFoldR } D (\vee (i :: is)) R (d, ds) &= \_, \_ \langle \$ \rangle^2 (\llbracket R \rrbracket d) \\
&\quad (\text{mapFoldR } D (\vee is) f ds) \\
\text{mapFoldR } D (\sigma S D') \quad R (s, ds) &= (\_, \_ s) \langle \$ \rangle (\text{mapFoldR } D (D' s) f ds)
\end{aligned}$$

**Figure 5.2** Definition of relational folds.

$$\llbracket - \rrbracket : \{I : \text{Set}\} \{D : \text{Desc } I\} \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \rightsquigarrow X) \rightarrow (\mu D \rightsquigarrow X)$$

The definition of  $\llbracket - \rrbracket$  is shown in Figure 5.2 (cf. the definition of *fold* in ??). For example, the relational fold on lists is, up to isomorphism,

$$\begin{aligned}
\llbracket - \rrbracket \{\top\} \{ListD A\} &: \{X : \top \rightarrow \text{Set}\} \rightarrow \\
&\quad (\mathbb{F} (ListD A) X \rightsquigarrow X) \rightarrow (\mu (ListD A) \rightsquigarrow X) \\
\llbracket R \rrbracket [] &= R ('nil, \blacksquare) \\
\llbracket R \rrbracket (a :: as) &= \llbracket R \rrbracket as \gg \lambda x \mapsto R ('cons, a, x, \blacksquare)
\end{aligned}$$

The functional and relational fold operators are related by the following lemma:

$$\begin{aligned}
\text{fun-preserves-fold} &: \{I : \text{Set}\} (D : \text{Desc } I) \{X : I \rightarrow \text{Set}\} \rightarrow \\
&\quad (f : \mathbb{F} D X \Rightarrow X) \{i : I\} (d : \mu D i) (x : X i) \rightarrow \\
&\quad \text{fun } (\text{fold } f) d x \cong \llbracket \text{fun } f \rrbracket d x
\end{aligned}$$

which is a strengthened version of  $\text{fun } (\text{fold } f) \simeq \llbracket \text{fun } f \rrbracket$ .

## 5.2 Definition of algebraic ornamentation

We now turn to algebraic ornamentation, the key construct that bridges internalist and relational programming, and look at a special case first. Let

$$R : \mathbb{F} (ListD A) (\text{const } X) \rightsquigarrow \text{const } X \quad \text{where } X : \text{Set}$$

**indexfirst data** AlgList  $A \ R : X \rightarrow \text{Set}$  **where**

There is an ornament from lists to algebraically ornamented lists which marks the fields  $rnil$ ,  $x'$ , and  $rcons$  in `AlgList` as additional and refines the index of the recursive position from  $\blacksquare$  to  $x'$ . The optimised predicate `(?)` for this ornament is

$$\begin{aligned} \text{AlgListP } A \ R \ x \ [] &\ni \text{nil } (rnil : R \ ('nil, \blacksquare) \ x) \\ \text{AlgListP } A \ R \ x \ (a :: as) &\ni \text{cons } (x' : X) \ (rcons : R \ ('cons, a, x', \blacksquare) \ x) \\ &\quad (p : \text{AlgListP } A \ R \ x' \ as) \end{aligned}$$
$$(x : X) \rightarrow \text{AlgList } A \ R \ x \cong \Sigma[as : \text{List } A] \ (\llbracket R \rrbracket as) \ x \quad (5.1)$$
$$\begin{aligned} length\text{-}alg &: \mathbb{F} (ListD\ A) (const\ Nat) \Rightarrow const\ Nat \\ length\text{-}alg &= ornAlg\ [NatD\text{-}ListD\ A] \end{aligned}$$
$$(n : \text{Nat}) \rightarrow \text{Vec}' A n \cong \Sigma[\text{as} : \text{List } A] \ (\langle \text{fun length-alg} \rangle) \text{ as } n$$
$$\text{Vec}' A n \cong \Sigma[as : \text{List } A] \text{ length } as \equiv n$$

by *fun-preserves-fold*.



$$\begin{aligned}
& \text{algROD} : \{I : \text{Set}\} (D : \text{RDesc } I) \{X : I \rightarrow \text{Set}\} \rightarrow \\
& \quad \mathcal{P} (\llbracket D \rrbracket X) \rightarrow \text{ROrnDesc } (\Sigma I X) \text{ outl } D \\
& \text{algROD } (\nu \text{ is}) \quad \{X\} P = \Delta[xs : \mathbb{P} \text{ is } X] \Delta[_ : P xs] \\
& \quad \nu (\mathbb{P}\text{-map } (\lambda \{i\} x \mapsto \text{ok } (i, x)) \text{ is } xs) \\
& \text{algROD } (\sigma S D) \quad P = \sigma[s : S] \text{ algROD } (D s) (\text{curry } P s) \\
& \text{algOD} : \{I : \text{Set}\} (D : \text{Desc } I) \{X : I \rightarrow \text{Set}\} \rightarrow \\
& \quad (\mathbb{F} D X \rightsquigarrow X) \rightarrow \text{OrnDesc } (\Sigma I X) \text{ outl } D \\
& \text{algOD } D R (\text{ok } (i, x)) = \text{algROD } (D i) ((R \circ) x)
\end{aligned}$$

**Figure 5.3** Definitions for algebraic ornamentation.

The above can be generalised to all datatypes encoded by the Desc universe. Let  $D : \text{Desc } I$  be a description and  $R : \mathbb{F} D X \rightsquigarrow X$  (where  $X : I \rightarrow \text{Set}$ ) an algebra. The **algebraic ornamentation** of  $D$  with  $R$  is an ornamental description

$$\text{algOD } D R : \text{OrnDesc } (\Sigma I X) \text{ outl } D$$

(where  $\text{outl} : \Sigma I X \rightarrow I$ ). The optimised predicate for  $\llbracket \text{algOD } D R \rrbracket$  is pointwise isomorphic to  $\llbracket R \rrbracket$ , i.e.,

$$(i : I) (x : X i) (d : \mu D i) \rightarrow \text{OptP } \llbracket \text{algOD } D R \rrbracket (\text{ok } (i, x)) d \cong \llbracket R \rrbracket d x$$

which is proved by induction on  $d$ . These isomorphisms give rise to a swap family

$$\text{algOD-FSwap } D R : \text{FSwap } (R\text{Sem } \llbracket \text{algOD } D R \rrbracket)$$

such that  $\text{Swap.P } (\text{algOD-FSwap } D R (\text{ok } (i, x))) = \lambda d \mapsto \llbracket R \rrbracket d x$ , so we arrive at the following conversion isomorphisms

$$(i : I) (x : X i) \rightarrow \mu \llbracket \text{algOD } D R \rrbracket (i, x) \cong \Sigma[d : \mu D i] \llbracket R \rrbracket d x \quad (5.2)$$

We get back AlgList by defining  $\text{AlgList } A R x = \mu \llbracket \text{algOD } (\text{List } D A) R \rrbracket (\blacksquare, x)$ . The definition of  $\text{algOD}$ , shown in Figure 5.3, is an adaptation and generalisation of McBride’s original definition of functional algebraic ornamentation [2011]. Roughly speaking, it retains (in the  $\sigma$  case of  $\text{algROD}$ ) all the fields of the base description and inserts (in the  $\nu$  case of  $\text{algROD}$ ) before every  $\nu$

- a new field of indices for the recursive positions (e.g., the field  $x'$  in AlgList) and

- another new field requesting a proof that
  - the indices supplied in the previous field and
  - the values for the fields originally in the base description
 compute to the targeted index through  $R$  (e.g., the fields *rnil* and *rcons* in AlgList).

Algebraic ornamentation is a convenient method for adding new indices to inductive families. (We will see in ?? that it is actually a canonical way to refine inductive families by ornamentation.) Most importantly, the conversion isomorphisms (5.2) state clearly what the new indices mean in terms of relations. We can thus easily translate relational expressions into internalist types for type-directed programming, as demonstrated in the next section.

## 5.3 Examples

We give three examples involving three relational theorems.

- Section 5.3.1 shows how the **Fold Fusion Theorem** [Bird and de Moor, 1997, Section 6.2] gives rise to conversion functions between algebraically ornamented datatypes.
- Section 5.3.2 implements the **Streaming Theorem** [Bird and Gibbons, 2003, Theorem 30] as an internalist program, whose type directly corresponds to the “metamorphic” specification stated by the theorem.
- Section 5.3.3 uses the **Greedy Theorem** [Bird and de Moor, 1997, Theorem 10.1] to nontrivially derive a suitable type for an internalist program that solves the **minimum coin change problem**.

### 5.3.1 The Fold Fusion Theorem

The statement of the **Fold Fusion Theorem** [Bird and de Moor, 1997, Section 6.2] is as follows: Let  $D : \text{Desc } I$  be a description,  $R : X \rightsquigarrow Y$  a relation, and  $S : \mathbb{F} D X \rightsquigarrow X$  and  $T : \mathbb{F} D Y \rightsquigarrow Y$  algebras. If  $R$  is a homomorphism from  $S$  to  $T$ , i.e.,

$$\begin{aligned}
& new\text{-}\Sigma : (I : \text{Set}) \{A : \text{Set}\} \{X : I \rightarrow \text{Set}\} \rightarrow \\
& \quad ((i : I) \rightarrow \text{Upgrade } A (X i)) \rightarrow \text{Upgrade } A (\Sigma I X) \\
& new\text{-}\Sigma I u = \mathbf{record} \{ P = \lambda a \mapsto \Sigma[i : I] \text{Upgrade}.P (u i) a \\
& \quad ; C = \lambda \{ a (i, x) \mapsto \text{Upgrade}.C (u i) a x \} \\
& \quad ; u = \lambda \{ a (i, p) \mapsto i, \text{Upgrade}.u (u i) a p \} \\
& \quad ; c = \lambda \{ a (i, p) \mapsto \text{Upgrade}.c (u i) a p \} \} \\
& \mathbf{syntax} \text{ new-}\Sigma I (\lambda i \rightarrow u) = \Sigma^+[i : I] u \\
& \_ \times^+ \_ : \{X Y : \text{Set}\} \rightarrow \text{Upgrade } X Y \rightarrow (Z : \text{Set}) \rightarrow \text{Upgrade } X (Y \times Z) \\
& u \times^+ Z = \mathbf{record} \{ P = \lambda x \mapsto \text{Upgrade}.P u x \times Z \\
& \quad ; C = \lambda \{ x (y, z) \mapsto \text{Upgrade}.C u x y \} \\
& \quad ; u = \lambda \{ x (p, z) \mapsto \text{Upgrade}.u u x p, z \} \\
& \quad ; c = \lambda \{ x (p, z) \mapsto \text{Upgrade}.c u x p \} \}
\end{aligned}$$

Figure 5.4 Two additional upgrade combinators.

$$R \cdot S \simeq T \cdot \mathbb{R} D R$$

which is referred to as the **fusion condition**, then we have

$$R \cdot \llbracket S \rrbracket \simeq \llbracket T \rrbracket$$

The above is, in fact, a corollary of two variations of Fold Fusion that replace relational equivalence in the statement of the theorem with relational inclusion. One variation is

$$R \cdot S \subseteq T \cdot \mathbb{R} D R \rightarrow R \cdot \llbracket S \rrbracket \subseteq \llbracket T \rrbracket \quad (5.3)$$

and the other variation simply reverses the direction of inclusion:

$$R \cdot S \supseteq T \cdot \mathbb{R} D R \rightarrow R \cdot \llbracket S \rrbracket \supseteq \llbracket T \rrbracket \quad (5.4)$$

Both of them roughly state that one fold can be conditionally transformed into another. Since algebraically ornamented datatypes are datatypes with constraints expressed as a fold, we should be able to transform these constraints by the Fold Fusion Theorem while leaving the underlying data unchanged, thus converting one algebraically ornamented datatype into another.

We look at (5.3) first. Assume that we have a proof of the antecedent

$$fcond_{\subseteq} : R \cdot S \subseteq T \cdot \mathbb{R} D R$$

Expanding the conclusion of (5.3) pointwise: if  $d : \mu D i$  folds to  $x : X i$  with  $S$ , which is “relaxed” to  $y : Y i$  by  $R$ , then  $d$  folds to  $y$  with  $T$ . This can be translated to a conversion function from a datatype algebraically ornamented with  $S$  to one with  $T$ :

$$\begin{aligned} fusion\text{-}conversion_{\subseteq} D R S T fcond_{\subseteq} : \\ \{i : I\} (x : X i) \rightarrow \mu [algOD D S] (i, x) \rightarrow \\ (y : Y i) \rightarrow R x y \rightarrow \mu [algOD D T] (i, y) \end{aligned}$$

This function does not alter the underlying  $(\mu D)$ -data, which can be easily expressed by upgrading (??) the identity function on  $\mu D$  to its type. We thus write the following upgrade

$$\begin{aligned} upg_{\subseteq} : \text{Upgrade } (\{i : I\} \rightarrow \mu D i \rightarrow \mu D i) \\ (\{i : I\} \{x : X i\} \rightarrow \mu [algOD D S] (i, x) \rightarrow \\ \{y : Y i\} \rightarrow R x y \rightarrow \mu [algOD D T] (i, y)) \\ upg_{\subseteq} = \forall[[i : I]] \forall^+[[x : X i]] \text{ref-}S i x \rightarrow \\ \forall^+[[y : Y i]] \forall^+[_ : R x y] \text{toUpgrade } (\text{ref-}T i y) \end{aligned}$$

where the refinements are defined by

$$\begin{aligned} \text{ref-}S : (i : I) (x : X i) \rightarrow \text{Refinement } (\mu D i) (\mu [algOD D S] (i, x)) \\ \text{ref-}S i x = \text{toRefinement } (algOD\text{-}FSwap D S (\text{ok } (i, x))) \\ \text{ref-}T : (i : I) (y : Y i) \rightarrow \text{Refinement } (\mu D i) (\mu [algOD D T] (i, y)) \\ \text{ref-}T i y = \text{toRefinement } (algOD\text{-}FSwap D T (\text{ok } (i, y))) \end{aligned}$$

and implement the conversion function by

$$\text{Upgrade.}u \text{ upg}_{\subseteq} id \{ \}_{0}$$

Goal 0 demands a promotion proof of type

$$\{i : I\} \{x : X i\} (d : \mu D i) \rightarrow ([S]) d x \rightarrow \{y : Y i\} \rightarrow R x y \rightarrow ([T]) (id d) y$$

which is exactly the pointwise expansion of the conclusion of (5.3). The coherence property

$$\begin{aligned} \{i : I\} \{x : X i\} (d : \mu D i) (d' : \mu [algOD D S] (i, x)) \rightarrow \\ \text{forget } [algOD D S] d' \equiv d \rightarrow \end{aligned}$$

$$\{y : Y\ i\} \rightarrow R\ x\ y \rightarrow \text{forget } [\text{algOD } D\ T] (\text{fusion-conversion}_{\subseteq} D\ R\ S\ T\ \text{fcond}_{\subseteq} d') \equiv \text{id } d$$

then states that the conversion function transforms the underlying  $(\mu D)$ -data in the same way as  $\text{id}$ , i.e., it leaves the underlying data unchanged. Similarly for (5.4), assuming that we have

$$\text{fcond}_{\supseteq} : R \cdot S \supseteq T \cdot \mathbb{R}\ D\ R$$

we should be able to construct the conversion function

$$\begin{aligned} &\text{fusion-conversion}_{\supseteq} D\ R\ S\ T\ \text{fcond}_{\supseteq} : \\ &\{i : I\} (y : Y\ i) \rightarrow \mu [\text{algOD } D\ T] (i, y) \rightarrow \\ &\Sigma[x : X\ i] \mu [\text{algOD } D\ S] (i, x) \times R\ x\ y \end{aligned}$$

as an upgraded version of the identity function with the upgrade

$$\begin{aligned} \text{upg}_{\supseteq} &: \text{Upgrade } (\{i : I\} \rightarrow \mu D\ i \rightarrow \mu D\ i) \\ &\quad (\{i : I\} \{y : Y\ i\} \rightarrow \mu [\text{algOD } D\ T] (i, y) \rightarrow \\ &\quad \Sigma[x : X\ i] \mu [\text{algOD } D\ S] (i, x) \times R\ x\ y) \\ \text{upg}_{\supseteq} &= \forall^+[[i : I]] \forall^+[[y : Y\ i]] \text{ref-T } i\ y \rightarrow \\ &\quad \Sigma^+[x : X\ i] \text{toUpgrade } (\text{ref-S } i\ x) \times^+ R\ x\ y \end{aligned}$$

in which we need two new combinators to deal with upgrading to product types, which are defined in Figure 5.4.

For a simple application, suppose that we need a “bounded” vector datatype, i.e., lists indexed with an upper bound on their length. A quick thought might lead to this definition

$$\begin{aligned} \text{BVec} &: \text{Set} \rightarrow \text{Nat} \rightarrow \text{Set} \\ \text{BVec } A\ m &= \mu [\text{algOD } (\text{ListD } A) (\text{geq} \cdot \text{fun length-alg})] (\blacksquare, m) \end{aligned}$$

where  $\text{geq} = \text{leq} \circ : \text{const Nat} \rightsquigarrow \text{const Nat}$  maps a natural number  $x$  to any natural number that is at least  $x$ . The conversion isomorphisms (5.2) specialise for  $\text{BVec}$  to

$$\text{BVec } A\ m \cong \Sigma[as : \text{List } A] ([\text{geq} \cdot \text{fun length-alg}])\ as\ m$$

for all  $m : \text{Nat}$ . But is  $\text{BVec}$  really the bounded vectors? Indeed it is, because we can deduce

$$\text{geq} \cdot ([\text{fun length-alg}]) \simeq ([\text{geq} \cdot \text{fun length-alg}])$$

by Fold Fusion. The fusion condition is

$$geq \cdot fun\ length\ alg \simeq geq \cdot fun\ length\ alg \cdot \mathbb{R}\ (ListD\ A)\ geq$$

The left-to-right inclusion is easily calculated as follows:

$$\begin{aligned} & geq \cdot fun\ length\ alg \\ \subseteq & \{ \text{identity} \} \\ & geq \cdot fun\ length\ alg \cdot idR \\ \subseteq & \{ \text{relator preserves identity} \} \\ & geq \cdot fun\ length\ alg \cdot \mathbb{R}\ (ListD\ A)\ idR \\ \subseteq & \{ geq\ \text{reflexive}; \text{relator is monotonic} \} \\ & geq \cdot fun\ length\ alg \cdot \mathbb{R}\ (ListD\ A)\ geq \end{aligned}$$

And from right to left:

$$\begin{aligned} & geq \cdot fun\ length\ alg \cdot \mathbb{R}\ (ListD\ A)\ geq \\ \subseteq & \{ fun\ length\ alg\ \text{monotonic on } geq \} \\ & geq \cdot geq \cdot fun\ length\ alg \\ \subseteq & \{ geq\ \text{transitive} \} \\ & geq \cdot fun\ length\ alg \end{aligned}$$

Note that these calculations are good illustrations of the power of relational calculation because of their simplicity — they are straightforward symbol manipulations, hiding details like quantifier reasoning behind the scenes. As demonstrated by the AoPA library [Mu et al., 2009], they can be faithfully formalised with preorder reasoning combinators in AGDA and used to discharge the fusion conditions of *fusion-conversion*<sub>⊆</sub> and *fusion-conversion*<sub>⊇</sub>. Hence we get two conversions, one of type

$$Vec\ A\ n \rightarrow (n \leq m) \rightarrow BVec\ A\ m$$

which relaxes a vector of length  $n$  to a bounded vector whose length is bounded above by some  $m$  that is at least  $n$ , and the other of type

$$BVec\ A\ m \rightarrow \Sigma[n : \text{Nat}] \ Vec\ A\ n \times (n \leq m)$$

which converts a bounded vector whose length is at most  $m$  to a vector of length precisely  $n$  and guarantees that  $n$  is at most  $m$ . Both conversions preserve the underlying list.

### 5.3.2 The Streaming Theorem for list metamorphisms

A **metamorphism** [Gibbons, 2007] is an unfold after a fold — it consumes a data structure to compute an intermediate value and then produces a new data structure using the intermediate value as the seed. In this section we will restrict ourselves to metamorphisms consuming and producing lists. As Gibbons noted, (list) metamorphisms in general cannot be automatically optimised in terms of time and space, but under certain conditions it is possible to refine a list metamorphism to a **streaming algorithm** — which can produce an initial segment of the output list without consuming all of the input list — or a parallel algorithm [Nakano, 2013]. In the rest of this section, we prove the **Streaming Theorem** [Bird and Gibbons, 2003, Theorem 30] by implementing the streaming algorithm given by the theorem with algebraically ornamented lists such that the algorithm satisfies its metamorphic specification by construction.

Our first step is to formulate a metamorphism as a relational specification of the streaming algorithm.

- The fold part needs a twist since using the conventional fold — known as the **right fold** for lists since the direction of computation on a list is from right to left (cf. wind direction) — does not easily give rise to a streaming algorithm. This is because we wish to talk about “partial consumption” naturally: for a list, partial consumption means examining and removing some elements of the list to get a sub-list on which we can resume consumption, and the natural way to do this is to consume the list from the left, examining and removing head elements and keeping the tail. We should thus use the **left fold** instead, which is usually defined as

$$\begin{aligned} foldl &: \{A\ X : \text{Set}\} \rightarrow (X \rightarrow A \rightarrow X) \rightarrow X \rightarrow \text{List } A \rightarrow X \\ foldl\ f\ x\ [] &= x \\ foldl\ f\ x\ (a :: as) &= foldl\ f\ (f\ x\ a)\ as \end{aligned}$$

The connection to the conventional fold (and thus algebraic ornamentation) is not lost, however — it is well known that a left fold can be alternatively implemented as a right fold by turning a list into a chain of functions of type  $X \rightarrow X$  transforming the initial value to the final result:

$$foldl\text{-}alg : \{A\ X : \text{Set}\} \rightarrow (X \rightarrow A \rightarrow X) \rightarrow$$

$$\begin{aligned}
& \mathbb{F} (ListD A) (const (X \rightarrow X)) \rightrightarrows const (X \rightarrow X) \\
& foldl\text{-}alg f ('nil, \quad \blacksquare) = id \\
& foldl\text{-}alg f ('cons, a, h, \blacksquare) = h \circ flip f a \\
& foldl : \{A X : Set\} \rightarrow (X \rightarrow A \rightarrow X) \rightarrow X \rightarrow List A \rightarrow X \\
& foldl f x as = fold (foldl\text{-}alg f) as x
\end{aligned}$$

The left fold can thus be linked to the relational fold by

$$fun (foldl f x) \simeq fun (\lambda h \mapsto h x) \cdot \llbracket fun (foldl\text{-}alg f) \rrbracket \quad (5.5)$$

- The unfold part is approximated by the converse of a relational fold: given a list coalgebra  $g : const X \rightrightarrows \mathbb{F} (ListD B) (const X)$  for some  $X : Set$ , we take its converse, turning it into a relational algebra, and use the converse of the relational fold with this algebra.

$$(\llbracket fun g^\circ \rrbracket)^\circ : const X \rightsquigarrow const (List A)$$

This is only an approximation because, while the relation does produce a list, the resulting list is inductive rather than coinductive, so the relation is actually a **well-founded** unfold, which is incapable of producing an infinite list.

Thus, given a “left algebra” for consuming List A

$$f : X \rightarrow A \rightarrow X$$

and a coalgebra for producing List B

$$g : const X \rightrightarrows \mathbb{F} (ListD B) (const X)$$

which together satisfy a **streaming condition** that we will see later, the streaming algorithm we implement, which takes as input the initial value  $x : X$  for the left fold, should be included in the following metamorphic relation:

$$meta f g x = (\llbracket fun g^\circ \rrbracket)^\circ \cdot fun (foldl f x) : const (List A) \rightsquigarrow const (List B)$$

Next we devise a type for the streaming algorithm that fully guarantees its correctness. By (5.5), the specification  $meta f g x$  is equivalent to

$$(\llbracket fun g^\circ \rrbracket)^\circ \cdot fun (\lambda h \mapsto h x) \cdot \llbracket fun (foldl\text{-}alg f) \rrbracket$$

Inspecting the above relation, we see that a conforming program takes a List A that folds to some  $h : X \rightarrow X$  with  $fun (foldl\text{-}alg f)$  and computes a List B that folds to  $h x : X$  with  $fun g^\circ$ . We thus implement the streaming algorithm by



$$\text{stream } f \ g : (x : X) \{h : X \rightarrow X\} \rightarrow \\ \text{AlgList } A \ (\text{fun } (\text{foldl-alg } f)) \ h \rightarrow \text{AlgList } B \ (\text{fun } g^\circ) \ (h \ x)$$

from which we can extract

$$\text{stream}' f \ g : X \rightarrow \text{List } A \rightarrow \text{List } B$$

which is guaranteed to satisfy

$$\text{fun } (\text{stream}' f \ g \ x) \subseteq \text{meta } f \ g \ x$$

The extraction of  $\text{stream}' f \ g$  from  $\text{stream } f \ g$  is done with the help of the conversion isomorphisms (5.2) for the two `AlgList` datatypes involved:

*consumption-iso* :

$$(h : X \rightarrow X) \rightarrow$$

$$\text{AlgList } A \ (\text{fun } (\text{foldl-alg } f)) \ h \cong \Sigma[as : \text{List } A] \ \text{fold } (\text{foldl-alg } f) \ as \equiv h$$

*production-iso* :

$$(x : X) \rightarrow \text{AlgList } B \ (\text{fun } g^\circ) \ x \cong \Sigma[bs : \text{List } B] \ (\llbracket \text{fun } g^\circ \rrbracket) \ bs \ x$$

(where *consumption-iso* has been simplified by *fun-preserves-fold*). Given  $x : X$ , what  $\text{stream}' f \ g \ x$  does is

- lifting the input  $as : \text{List } A$  to an algebraic list of type

$$\text{AlgList } A \ (\text{fun } (\text{foldl-alg } f)) \ (\text{fold } (\text{foldl-alg } f) \ as)$$

using the right-to-left direction of *consumption-iso*  $(\text{fold } (\text{foldl-alg } f) \ as)$  (with the equality proof obligation discharged trivially by *refl*),

- transforming this algebraic list to a new one of type

$$\text{AlgList } B \ (\text{fun } g^\circ) \ (\text{foldl } f \ x \ as)$$

using  $\text{stream } f \ g \ x$ , and

- demoting the new algebraic list to `List B` using the left-to-right direction of *production-iso*  $(\text{foldl } f \ x \ as)$ .

The use of *production-iso* in the last step ensures that the result  $\text{stream}' f \ g \ x \ as : \text{List } B$  satisfies

$$(\llbracket \text{fun } g^\circ \rrbracket) (\text{stream}' f \ g \ x \ as) (\text{foldl } f \ x \ as)$$

which easily implies

$$((\llbracket \text{fun } g^\circ \rrbracket)^\circ \cdot \text{fun } (\text{foldl } f \ x)) \ as \ (\text{stream}' f \ g \ x \ as)$$

i.e.,  $\text{fun } (\text{stream}' f g x) \subseteq \text{meta } f g x$ , as required.

What is left is the implementation of  $stream\ f\ g$ . Operationally, we maintain a state of type  $X$  (and hence require an initial state as an input to the function), and we can try either


- to update the state by consuming elements of  $A$  with  $f$ , or
- to produce elements of  $B$  (and transit to a new state) by applying  $g$  to the state.

Since we want *stream f g* to be as productive as possible, we should always try to produce elements of *B* with *g* first, and only try to consume elements of *A* with *f* when *g* produces nothing. In AGDA:

$$\begin{array}{l}
stream\ f\ g : (x : X) \{h : X \rightarrow X\} \rightarrow \\
\quad AlgList\ A\ (fun\ (foldl\_alg\ f))\ h \rightarrow AlgList\ B\ (fun\ g^\circ)\ (h\ x) \\
\\
stream\ f\ g\ x \quad as \qquad \qquad \qquad \textbf{with}\ g\ x \quad | \quad inspect\ g\ x \\
stream\ f\ g\ x\ \{h\}\ as \quad \quad \quad | \quad next\ b\ x' \quad | \quad [gxeq] = cons\ b\ (h\ x')\ \{\}_{0} \\
\qquad \qquad \qquad \qquad \qquad \qquad \qquad \qquad \qquad \qquad \qquad (stream\ f\ g\ x'\ as) \\
\\
stream\ f\ g\ x \quad (nil \quad refl) \quad | \quad nothing \quad | \quad [gxeq] = nil\ gxeq \\
stream\ f\ g\ x \quad (cons\ a\ h'\ refl\ as) \quad | \quad nothing \quad | \quad [gxeq] = stream\ f\ g\ (f\ x\ a)\ as
\end{array}$$

We match  $g\ x$  with either of the two patterns next  $b\ x' = ('cons', b, x', \blacksquare)$  and  $nothing = ('nil', \blacksquare)$ .

- If the result is next  $b\ x'$ , we should emit  $b$  and use  $x'$  as the new state; the recursively computed algebraic list is indexed with  $h\ x'$ , and we are left with a proof obligation of type  $g\ (h\ x) \equiv \text{next } b\ (h\ x')$  at Goal 0; we will come back to this proof obligation later.
- If the result is nothing, we should attempt to consume from the input list.
  - If the input list is empty, implying that the index  $h$  of its type is just  $id$ , both production and consumption have ended, so we return an empty list. The `nil` constructor requires a proof of  $(fun\ g\ \circ)\ \text{nothing}\ (h\ x)$ , which reduces to  $g\ x \equiv \text{nothing}$  and is discharged with the help of the “inspect idiom” in AGDA’s standard library (which, in a **with**-matching, gives a proof that the term being matched (in this case  $g\ x$ ) is propositionally equal to the matched pattern (in this case `nothing`)).
  - Otherwise the input list is nonempty, implying that  $h$  is  $h' \circ flip\ f\ a$  where



**Figure 5.5** State transitions involved in commutativity of production and consumption (cf. Gibbons [2007, Figures 1 and 2]).

$a$  is the head of the input list, and we should continue with the new state  $f\ x\ a$ , keeping the tail for further consumption. Typing directly works out because the index of the recursive result  $h'\ (f\ x\ a)$  and the required index  $(h' \circ \text{flip}\ f\ a)\ x$  are definitionally equal.

Now we look at Goal 0. We have

$$gxeq : g\ x \equiv \text{next}\ b\ x'$$

in the context, and need to prove

$$g\ (h\ x) \equiv \text{next}\ b\ (h\ x')$$

This is commutativity of production and consumption (see Figure 5.5): The function  $h : X \rightarrow X$  is the state transformation resulting from consumption of the input list  $as$ . From the initial state  $x$ , we can either

- apply  $g$  to  $x$  to produce  $b$  and reach a new state  $x'$ , and then apply  $h$  to consume the list and update the state to  $h\ x'$ , or
- apply  $h$  to consume the list and update the state to  $h\ x$ , and then apply  $g$  to  $h\ x$  to produce an element and reach a new state,

and we need to prove that the outcomes are the same: doing production using  $g$  and consumption using  $h$  in whichever order should emit the same element and reach the same final state. This cannot be true in general, so we should impose some commutativity condition on  $f$  and  $g$ , which is called the **streaming condition**:

$$\text{StreamingCondition}\ f\ g : \text{Set}$$

*StreamingCondition*  $f\ g =$

$$(a : A) (b : B) (x\ x' : X) \rightarrow g\ x \equiv \text{next } b\ x' \rightarrow g\ (f\ x\ a) \equiv \text{next } b\ (f\ x'\ a)$$

The streaming condition is commutativity of one step of production and consumption, whereas the proof obligation at Goal 0 is commutativity of one step of production and multiple steps of consumption (of the entire list), so we perform a straightforward induction to extend the streaming condition along the axis of consumption:

*streaming-lemma* :

$$(b : B) (x\ x' : X) \rightarrow g\ x \equiv \text{next } b\ x' \rightarrow$$

$$\{h : X \rightarrow X\} \rightarrow \text{AlgList } A\ (\text{fun } (f\text{oldl-}alg\ f))\ h \rightarrow g\ (h\ x) \equiv \text{next } b\ (h\ x')$$

$$\text{streaming-lemma } b\ x\ x'\ eq\ (\text{nil} \quad \text{refl}) = eq$$

$$\text{streaming-lemma } b\ x\ x'\ eq\ (\text{cons } a\ h\ \text{refl } as) =$$

$$\text{streaming-lemma } b\ (f\ x\ a)\ (f\ x'\ a)\ (\text{streaming-condition } f\ g\ a\ b\ x\ x'\ eq)\ as$$

where *streaming-condition* : *StreamingCondition*  $f\ g$  is a proof term that should be supplied along with  $f$  and  $g$  in the beginning. Goal 0 is then discharged by the term *streaming-lemma*  $b\ x\ x'\ g\ x\ eq\ as$ .

We have thus completed the implementation of the Streaming Theorem, except that *stream*  $f\ g$  is non-terminating, as there is no guarantee that  $g$  produces only a finite number of elements. In our setting, where the output list is specified to be finite, we can additionally require that  $g$  is well-founded and revise *stream* accordingly (see, e.g., Nordström [1988]); the general way out is to switch to coinductive datatypes to allow the output list to be infinite, which, however, falls outside the scope of this dissertation.

It is interesting to compare our implementation with the proofs of Bird and Gibbons [2003]. While their Lemma 29 turns explicitly into our *streaming-lemma*, their Theorem 30 goes implicitly into the typing of *stream* and no longer needs special attention. The structure of *stream* already matches that of Bird and Gibbons's proof of their Theorem 30, and the principled type design using algebraic ornamentation elegantly loads the proof onto the structure of *stream* — this is internalism at its best.

### 5.3.3 The Greedy Theorem and the minimum coin change problem

Suppose that we have an unlimited number of 1-penny, 2-pence, and 5-pence coins, modelled by the following datatype:

**data** Coin : Set **where**

1p : Coin

2p : Coin

5p : Coin

Given  $n : \text{Nat}$ , the **minimum coin change problem** asks for the least number of coins that make up  $n$  pence. We can give a relational specification of the problem with the following “minimisation” operator:

$$\begin{aligned} \min\_ \bullet \Lambda\_ : \{I : \text{Set}\} \{X Y : I \rightarrow \text{Set}\} (R : Y \rightsquigarrow Y) (S : X \rightsquigarrow Y) &\rightarrow (X \rightsquigarrow Y) \\ \min\_ \bullet \Lambda\_ \{Y := Y\} R S \{i\} x y &= S x y \times ((y' : Y i) \rightarrow S x y' \rightarrow R y' y) \end{aligned}$$

An input  $x : X i$  for some  $i : I$  is mapped by  $\min R \bullet \Lambda S$  to  $y : Y i$  if  $y$  is a possible result in  $S x : \mathcal{P}(Y i)$  and is the smallest such result under  $R$ , in the sense that any  $y'$  in  $S x : \mathcal{P}(Y i)$  must satisfy  $R y' y$ . (We think of  $R$  as mapping larger inputs to smaller outputs.) Intuitively, we can think of  $\min R \bullet \Lambda S$  as consisting of two steps: the first step  $\Lambda S$  computes the set of all possible results yielded by  $S$ , and the second step  $\min R$  nondeterministically chooses a minimum result from that set. We use bags of coins as the type of solutions, and represent them as decreasingly ordered lists indexed with an upper bound. (This is a deliberate choice to make the derivation work, but one would naturally be led to this representation having attempted to apply the **Greedy Theorem**, which will be introduced shortly.) If we define the ordering on coins as

$$\_ \leqslant_{\text{C}} \_ : \text{Coin} \rightarrow \text{Coin} \rightarrow \text{Set}$$

$$c \leqslant_{\text{C}} d = \text{value } c \leqslant \text{value } d$$

where the values of the coins are defined by

$$\text{value} : \text{Coin} \rightarrow \text{Nat}$$

$$\text{value } 1\text{p} = 1$$

$$\text{value } 2\text{p} = 2$$

$$\text{value } 5\text{p} = 5$$

then the datatype of coin bags we use is

```

CoinBagOD : OrnDesc Coin ! (ListD Coin)
CoinBagOD = OrdListOD Coin (flip _≤C_)
-- specialising Val to Coin and _≤_ to flip _≤C_
indexfirst data CoinBag : Coin → Set where
  CoinBag c ⊃ nil
    | cons (d : Coin) (leq : d ≤C c) (b : CoinBag d)

```

The total value of a coin bag is the sum of the values of the coins in the bag, which is computed by a (functional) fold:

```

total-value-alg : IF [CoinBagOD] (const Nat) ⇒ const Nat
total-value-alg ('nil , ■) = 0
total-value-alg ('cons , d , _ , n , ■) = value d + n
total-value : CoinBag ⇒ const Nat
total-value = fold total-value-alg

```

and the number of coins in a coin bag is computed by an ornamental forgetful function shrinking ordered lists to natural numbers:

```

size-alg : IF [CoinBagOD] (const Nat) ⇒ const Nat
size-alg = ornAlg (NatD-ListD Coin ⊙ [CoinBagOD])
size : CoinBag ⇒ const Nat
size = fold size-alg
-- which is definitionally forget (NatD-ListD Coin ⊙ [CoinBagOD])

```

The specification of the minimum coin change problem can now be written as

```

min-coin-change : const Nat ⇔ CoinBag
min-coin-change = min (fun sizeo • leq • fun size) • Λ (fun total-valueo)

```

Intuitively, given an input  $n : \text{Nat}$ , the relation  $\text{fun total-value}^o$  computes an arbitrary coin bag whose total value is  $n$ , so  $\text{min-coin-change}$  first computes the set of all such coin bags and then chooses from the set a coin bag whose size is smallest. Our goal, then, is to write a functional program  $f : \text{const Nat} \Rightarrow \text{CoinBag}$  such that  $\text{fun } f \subseteq \text{min-coin-change}$ , and then  $f \{5p\} : \text{Nat} \rightarrow \text{CoinBag } 5p$  would be a solution. (The type  $\text{CoinBag } 5p$  contains all coin bags, since  $5p$  is the largest denomination and hence a trivial upper bound on the content of bags.) Of

course, we may guess what  $f$  should look like, but its correctness proof is much harder. Can we construct the program and its correctness proof in a more manageable way?

### The plan

In traditional relational program derivation [Bird and de Moor, 1997], we would attempt to refine the specification *min-coin-change* to some simpler relational program and then to an executable functional program by applying algebraic laws and theorems. With algebraic ornamentation, however, there is a new possibility: if, for some algebra  $R : \mathbb{F} \lfloor \text{CoinBagOD} \rfloor (\text{const Nat}) \rightsquigarrow \text{const Nat}$ , we can derive

$$(\lfloor R \rfloor)^\circ \subseteq \text{min-coin-change} \quad (5.6)$$

then we can manufacture a new datatype

$\text{GreedyBagOD} : \text{OrnDesc} (\text{Coin} \times \text{Nat}) \text{ outl } \lfloor \text{CoinBagOD} \rfloor$

$\text{GreedyBagOD} = \text{algOD } \lfloor \text{CoinBagOD} \rfloor R$

$\text{GreedyBag} : \text{Coin} \rightarrow \text{Nat} \rightarrow \text{Set}$

$\text{GreedyBag } c \ n = \mu \lfloor \text{GreedyBagOD} \rfloor (c, n)$

and construct a function of type

$\text{greedy} : (c : \text{Coin}) (n : \text{Nat}) \rightarrow \text{GreedyBag } c \ n$

from which we can assemble a solution

$\text{sol} : \text{Nat} \rightarrow \text{CoinBag } 5p$

$\text{sol} = \text{forget } \lceil \text{GreedyBagOD} \rceil \circ \text{greedy } 5p$

The program  $\text{sol}$  satisfies the specification because of the following argument:

For any  $c : \text{Coin}$  and  $n : \text{Nat}$ , by (5.2) we have

$$\text{GreedyBag } c \ n \cong \Sigma [b : \text{CoinBag } c] (\lfloor R \rfloor) b \ n$$

In particular, since the first half of the left-to-right direction of the isomorphism is  $\text{forget } \lceil \text{GreedyBagOD} \rceil$ , we have

$$(\lfloor R \rfloor) (\text{forget } \lceil \text{GreedyBagOD} \rceil g) n$$

for any  $g : \text{GreedyBag } c \ n$ . Substituting  $g$  by  $\text{greedy } 5p \ n$ , we get

$$(\llbracket R \rrbracket) (sol\ n)\ n$$

which implies, by (5.6),

$$min\text{-}coin\text{-}change\ n\ (sol\ n)$$

i.e., *sol* satisfies the specification. Thus all we need to do to solve the minimum coin change problem is

- refine the specification *min-coin-change* to the converse of a fold, i.e., find the algebra *R* in (5.6), and
- construct the internalist program *greedy*.

### Refining the specification

The key to refining *min-coin-change* to the converse of a fold lies in the following version of the **Greedy Theorem**, which is a specialisation of Bird and de Moor's Theorem 10.1 modulo indexing: Let  $D : \text{Desc } I$  be a description,  $R : \mu D \rightsquigarrow \mu D$  a preorder, and  $S : \mathbb{F} D\ X \rightsquigarrow X$  an algebra. Consider the specification

$$min\ R \cdot \Lambda (\llbracket S \rrbracket^\circ)$$

That is, given an input value  $x : X\ i$  for some  $i : I$ , we choose a minimum under *R* among all those elements of  $\mu D\ i$  that computes to *x* through  $\llbracket S \rrbracket$ . The Greedy Theorem states that, if the initial algebra

$$\alpha = fun\ con : \mathbb{F} D (\mu D) \rightsquigarrow \mu D$$

is monotonic on *R*, i.e.,

$$\alpha \cdot \mathbb{R} D\ R \subseteq R \cdot \alpha$$

and there is a relation (ordering)  $Q : \mathbb{F} D\ X \rightsquigarrow \mathbb{F} D\ X$  such that the **greedy condition**

$$\alpha \cdot \mathbb{R} D (\llbracket S \rrbracket^\circ) \cdot (Q \cap (S^\circ \cdot S))^\circ \subseteq R^\circ \cdot \alpha \cdot \mathbb{R} D (\llbracket S \rrbracket^\circ)$$

is satisfied, then we have

$$\llbracket (min\ Q \cdot \Lambda (S^\circ))^\circ \rrbracket^\circ \subseteq min\ R \cdot \Lambda (\llbracket S \rrbracket^\circ)$$

The Greedy Theorem essentially reduces a global optimisation problem (as indicated by the outermost *min R*) to a local optimisation problem (as indicated



by the *min Q* inside the relational fold). The theorem admits an elegant calculational proof, which can be faithfully reprised in AGDA. Here we offer an intuitive explanation: The monotonicity condition states that if  $ds : \mathbb{F} D (\mu D) i$  for some  $i : I$  is better than  $ds' : \mathbb{F} D (\mu D) i$  under  $\mathbb{R} D R$ , i.e.,  $ds$  and  $ds'$  are equal except that the recursive positions of  $ds$  are all better than the corresponding recursive positions of  $ds'$  under  $R$ , then  $\text{con } ds : \mu D i$  would be better than  $\text{con } ds' : \mu D i$  under  $R$ . This implies that, when solving the optimisation problem, better solutions to subproblems would lead to a better solution to the original problem, so the **principle of optimality** applies — to reach an optimal solution, it suffices to find optimal solutions to subproblems, and we are entitled to use the converse of a fold to find optimal solutions recursively. The greedy condition further states that there is an ordering  $Q$  on the ways of decomposing the problem that has significant influence on the quality of solutions: Suppose that there are two decompositions  $xs$  and  $xs' : \mathbb{F} D X i$  of some problem  $x : X i$  for some  $i : I$ , i.e., both  $xs$  and  $xs'$  are in  $S^\circ x : \mathcal{P}(\mathbb{F} D X i)$ , and assume that  $xs$  is better than  $xs'$  under  $Q$ . Then for any solution resulting from  $xs'$  (computed by  $\alpha \cdot \mathbb{R} D ((\lfloor S \rfloor)^\circ)$ ) there always exists a better solution resulting from  $xs$ , so ignoring  $xs'$  would only rule out worse solutions. The greedy condition thus guarantees that we will arrive at an optimal solution by always choosing the best decomposition, which is done by  $\text{min } Q \cdot \Lambda (S^\circ) : X \rightsquigarrow \mathbb{F} D X$ .

Back to the minimum coin change problem. By *fun-preserves-fold*, the specification *min-coin-change* is equivalent to

$$\text{min } (\text{fun size}^\circ \cdot \text{leq} \cdot \text{fun size}) \cdot \Lambda ((\lfloor \text{fun total-value-alg} \rfloor)^\circ)$$

which matches the form of the generic specification given in the Greedy Theorem, so we try to discharge the two conditions of the theorem. The monotonicity condition reduces to monotonicity of *fun size-alg* on *leq*, and can be easily proved either by relational calculation or pointwise reasoning. As for the greedy condition, an obvious choice for  $Q$  is an ordering that leads us to choose the largest possible denomination, so we go for

$$\begin{aligned} Q : \mathbb{F} [\text{CoinBagOD}] (\text{const Nat}) &\rightsquigarrow \mathbb{F} [\text{CoinBagOD}] (\text{const Nat}) \\ Q ('nil \quad , \quad \blacksquare) &= \text{return } ('nil \quad , \quad \blacksquare) \\ Q ('cons \quad , \quad d \quad , \quad -) &= (\lambda e \text{ rest} \mapsto 'cons \quad , \quad e \quad , \quad \text{rest}) \prec_{\$}^2 (-\leq_{\text{C}} d) \text{ any} \end{aligned}$$

```

data CoinOrderedView : Coin → Coin → Set where
  1p1p : CoinOrderedView 1p 1p
  1p2p : CoinOrderedView 1p 2p
  1p5p : CoinOrderedView 1p 5p
  2p2p : CoinOrderedView 2p 2p
  2p5p : CoinOrderedView 2p 5p
  5p5p : CoinOrderedView 5p 5p

view-ordered-coin : (c d : Coin) → c ≤C d → CoinOrderedView c d

data CoinBag'View : {c : Coin} {n : Nat} {l : Nat} → CoinBag' c n l → Set where
  empty : {c : Coin} → CoinBag'View {c} {0} {0} bnul
  1p1p : {m l : Nat} {lep : 1p ≤C 1p}
    (b : CoinBag' 1p m l) → CoinBag'View {1p} {1 + m} {1 + l} (bcons 1p lep b)
  1p2p : {m l : Nat} {lep : 1p ≤C 2p}
    (b : CoinBag' 1p m l) → CoinBag'View {2p} {1 + m} {1 + l} (bcons 1p lep b)
  2p2p : {m l : Nat} {lep : 2p ≤C 2p}
    (b : CoinBag' 2p m l) → CoinBag'View {2p} {2 + m} {1 + l} (bcons 2p lep b)
  1p5p : {m l : Nat} {lep : 1p ≤C 5p}
    (b : CoinBag' 1p m l) → CoinBag'View {5p} {1 + m} {1 + l} (bcons 1p lep b)
  2p5p : {m l : Nat} {lep : 2p ≤C 5p}
    (b : CoinBag' 2p m l) → CoinBag'View {5p} {2 + m} {1 + l} (bcons 2p lep b)
  5p5p : {m l : Nat} {lep : 5p ≤C 5p}
    (b : CoinBag' 5p m l) → CoinBag'View {5p} {5 + m} {1 + l} (bcons 5p lep b)

view-CoinBag' : {c : Coin} {n l : Nat} (b : CoinBag' c n l) → CoinBag'View b

```

**Figure 5.6** Two views for proving *greedy-lemma*.

$\text{greedy-lemma} : (c\ d : \text{Coin}) \rightarrow c \leqslant d \rightarrow (m\ n : \text{Nat}) \rightarrow \text{value } c + m \equiv \text{value } d + n \rightarrow$	
$(l : \text{Nat}) (b : \text{CoinBag}'\ c\ m\ l) \rightarrow \Sigma[l' : \text{Nat}] \text{ CoinBag}'\ d\ n\ l' \times (l' \leqslant l)$	
$\text{greedy-lemma}\ c\ d\ c \leqslant d\ m\ n\ eq\ l\ b\ \textbf{with}\ view\text{-ordered-coin}\ c\ d\ c \leqslant d$	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 1p\ n\ l' \times (l' \leqslant l)\}_0$
$\text{greedy-lemma}\ 1.p\ 1.p - .n\ refl\ l\ b_{\text{CoinBag}'\ 1p\ n\ l}$	$1p1p = \{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 1p\ n\ l' \times (l' \leqslant l)\}_0$
$\text{greedy-lemma}\ 1.p\ 2.p - .(1+n)\ n\ refl\ l\ b$	$b \mid 1p2p\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 1.p\ 2.p - .(1+n)\ n\ refl.\ (1+l'')\ .-$	$1p2p \mid 1p1p\ \{n\}\ \{l''\}\ b\ \text{CoinBag}'\ 1p\ n\ l'' =$
	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 2p\ n\ l' \times (l' \leqslant 1 + l'')\}_1$
$\text{greedy-lemma}\ 1.p\ 5.p - .(4+n)\ n\ refl\ l\ b$	$b \mid 1p5p\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 1.p\ 5.p - .(4+n)\ n\ refl.\ .-$	$1p5p \mid 1p1p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 1.p\ 5.p - .(4+n)\ n\ refl.\ .-$	$1p5p \mid 1p1p\ .-\mid 1p1p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 1.p\ 5.p - .(4+n)\ n\ refl.\ .-$	$1p5p \mid 1p1p\ .-\mid 1p1p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 1.p\ 5.p - .(4+n)\ n\ refl.\ (4+l'')\ .-$	$1p5p \mid 1p1p\ .-\mid 1p1p\ .-\mid 1p1p\ \{n\}\ \{l''\}\ b\ \text{CoinBag}'\ 1p\ n\ l'' =$
	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 5p\ n\ l' \times (l' \leqslant 4 + l'')\}_2$
$\text{greedy-lemma}\ 2.p\ 2.p - .n\ refl\ l\ b_{\text{CoinBag}'\ 2p\ n\ l}$	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 2p\ n\ l' \times (l' \leqslant l)\}_3$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl\ l\ b$	$b \mid 2p5p\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl.\ .-$	$2p5p \mid 1p2p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl.\ .-$	$2p5p \mid 1p2p\ .-\mid 1p1p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl.\ (3+l'')\ .-$	$2p5p \mid 1p2p\ .-\mid 1p1p\ .-\mid 1p1p\ \{n\}\ \{l''\}\ b\ \text{CoinBag}'\ 1p\ n\ l'' =$
	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 5p\ n\ l' \times (l' \leqslant 3 + l'')\}_4$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl.\ .-$	$2p5p \mid 2p2p\ b\ \textbf{with}\ view\text{-CoinBag}'\ b$
$\text{greedy-lemma}\ 2.p\ 5.p - .(3+n)\ n\ refl.\ (2+l'')\ .-$	$2p5p \mid 2p2p\ .-\mid 1p2p\ \{n\}\ \{l''\}\ b\ \text{CoinBag}'\ 2p\ n\ l'' =$
	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 5p\ n\ l' \times (l' \leqslant 2 + l'')\}_5$
$\text{greedy-lemma}\ 2.p\ 5.p - .(4+k).\ (1+k)\ refl.\ (2+l'')\ .-$	$2p5p \mid 2p2p\ .-\mid 2p2p\ \{k\}\ \{l''\}\ b\ \text{CoinBag}'\ 2p\ k\ l'' =$
	$\{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 5p\ (1+k)\ l' \times (l' \leqslant 2 + l'')\}_6$
$\text{greedy-lemma}\ 5.p\ 5.p - .n\ refl\ l\ b_{\text{CoinBag}'\ 5p\ n\ l}$	$5p5p = \{\Sigma[l' : \text{Nat}] \text{ CoinBag}'\ 5p\ n\ l' \times (l' \leqslant l)\}_7$

**Figure 5.7** Cases of *greedy-lemma*, generated semi-automatically by AGDA’s interactive case-split mechanism. Goal types are shown in the interaction points, and the types of some pattern variables are shown in subscript beside them.

where, in the cons case, the output is required to be also a cons node, and the coin at its head position must be one that is no smaller than the coin  $d$  at the head position of the input. It is nontrivial to prove the greedy condition by relational calculation. Here we offer instead a brute-force yet conveniently expressed case analysis by pattern matching. Define a new datatype  $\text{CoinBag}'$  by composing two algebraic ornaments on  $\lfloor \text{CoinBagOD} \rfloor$  in parallel:

$$\begin{aligned} \text{CoinBag}'\text{OD} &: \text{OrnDesc } (\text{outl} \bowtie \text{outl}) \text{ pull } \lfloor \text{CoinBagOD} \rfloor \\ \text{CoinBag}'\text{OD} &= \lceil \text{algOD } \lfloor \text{CoinBagOD} \rfloor \text{ (fun total-value-alg)} \rceil \otimes \\ &\quad \lceil \text{algOD } \lfloor \text{CoinBagOD} \rfloor \text{ (fun size-alg)} \rceil \\ \text{CoinBag}' &: \text{Coin} \rightarrow \text{Nat} \rightarrow \text{Nat} \rightarrow \text{Set} \\ \text{CoinBag}' &= \mu \lfloor \text{CoinBag}'\text{OD} \rfloor (\text{ok } (c, n), \text{ok } (c, l)) \end{aligned}$$

whose two constructors can be specialised to

$$\begin{aligned} \text{bnil} &: \{c : \text{Coin}\} \rightarrow \text{CoinBag}' c 0 0 \\ \text{bcons} &: \{c : \text{Coin}\} \{n l : \text{Nat}\} \rightarrow (d : \text{Coin}) \rightarrow d \leq_c c \rightarrow \\ &\quad \text{CoinBag}' d n l \rightarrow \text{CoinBag}' c (\text{value } d + n) (1 + l) \end{aligned}$$

By predicate swapping using the modularity isomorphisms (??) and *fun-preserves-fold*,  $\text{CoinBag}'$  is characterised by the isomorphisms

$$\text{CoinBag}' c n l \cong \Sigma [b : \text{CoinBag } c] (\text{total-value } b \equiv n) \times (\text{size } b \equiv l) \quad (5.7)$$

for all  $c : \text{Coin}$ ,  $n : \text{Nat}$ , and  $l : \text{Nat}$ . Hence a coin bag of type  $\text{CoinBag}' c n l$  contains  $l$  coins that are no larger than  $c$  and sum up to  $n$  pence. The greedy condition then essentially reduces to this lemma:

$$\begin{aligned} \text{greedy-lemma} &: (c d : \text{Coin}) \rightarrow c \leq_c d \rightarrow \\ &\quad (m n : \text{Nat}) \rightarrow \text{value } c + m \equiv \text{value } d + n \rightarrow \\ &\quad (l : \text{Nat}) (b : \text{CoinBag}' c m l) \rightarrow \\ &\quad \Sigma [l' : \text{Nat}] \text{CoinBag}' d n l' \times (l' \leq l) \end{aligned}$$

That is, given a problem (i.e., a value to be represented by coins), if  $c : \text{Coin}$  is a choice of decomposition (i.e., the first coin used) no better than  $d : \text{Coin}$  (i.e.,  $c \leq_c d$  — recall that we prefer larger denominations), and  $b : \text{CoinBag}' c m l$  is a solution of size  $l$  to the remaining subproblem  $m$  resulting from choosing  $c$ , then there is a solution to the remaining subproblem  $n$  resulting from choosing  $d$  whose size  $l'$  is no greater than  $l$ . We define two views (??) to aid the analysis, whose datatypes and covering functions are shown in Figure 5.6:

- the first view analyses a proof of  $c \leq_C d$  and exhausts all possibilities of  $c$  and  $d$ , and
- the second view analyses some  $b : \text{CoinBag}' c n l$  and exhausts all possibilities of  $c$ ,  $n$ ,  $l$ , and the first coin in  $b$  (if any).

The function *greedy-lemma* can then be split into eight cases by first exhausting all possibilities of  $c$  and  $d$  by the first view and then analysing the content of  $b$  by the second view. Figure 5.7 shows the case-split tree generated semi-automatically by AGDA; the detail is explained as follows:

- At Goal 0 (and similarly Goals 3 and 7), the input bag is  $b : \text{CoinBag}' 1p n l$ , and we should produce a  $\text{CoinBag}' 1p n l'$  for some  $l' : \text{Nat}$  such that  $l' \leq l$ . This is easy because  $b$  itself is a suitable bag.
- At Goal 1 (and similarly Goals 2, 4, and 5), the input bag has type  $\text{CoinBag}' 1p (1 + n) l$ , i.e., the coins in the bag are no larger than 1p and the total value is  $1 + n$ . The bag must contain 1p as its first coin; let the rest of the bag be  $b : \text{CoinBag}' 1p n l''$ . At this point AGDA can deduce that  $l$  must be  $1 + l''$ . Now we can return  $b$  as the result after the upper bound on its coins is relaxed from 1p to 2p, which is done by

$$\text{relax} : \{c d : \text{Coin}\} \{n l : \text{Nat}\} \rightarrow c \leq_C d \rightarrow \text{CoinBag}' c n l \rightarrow \text{CoinBag}' d n l$$

- The remaining Goal 6 is the most interesting one: The input bag has type  $\text{CoinBag}' 2p (3 + n) l$ , which in this case contains two 2-pence coins, and the rest of the bag is  $b : \text{CoinBag}' 2p k l''$ . AGDA deduces that  $n$  must be  $1 + k$  and  $l$  must be  $2 + l''$ . We thus need to add a penny to  $b$  to increase its total value to  $1 + k$ , which is done by

$$\begin{aligned} \text{add-penny} : \\ \{c : \text{Coin}\} \{n l : \text{Nat}\} \rightarrow \text{CoinBag}' c n l \rightarrow \text{CoinBag}' c (1 + n) (1 + l) \end{aligned}$$

and relax the bound of *add-penny*  $b$  from 2p to 5p.

The above case analysis may look tedious, but AGDA is able to

- produce all the cases (modulo some cosmetic revisions) after the programmer decides to use the two views and instructs AGDA to do case splitting accordingly, and
- manage all the bookkeeping and deductions about the total value and the

size of bags with dependent pattern matching,  
 so the overhead on the programmer's side is actually less than it seems. The greedy condition can now be discharged by pointwise reasoning, using (5.7) to interface with *greedy-lemma*. We conclude that the Greedy Theorem is applicable, and obtain

$$(\llbracket (\min Q \cdot \Lambda (\text{fun total-value-alg } \circ)) \circ \rrbracket)^\circ \subseteq \text{min-coin-change}$$

We have thus found the algebra

$$R = (\min Q \cdot \Lambda (\text{fun total-value-alg } \circ))^\circ$$

which will help us to construct the final internalist program.

### Constructing the internalist program

As planned, we synthesise a new datatype by ornamenting CoinBag using the algebra  $R$  derived above:

$\text{GreedyBagOD} : \text{OrnDesc } (\text{Coin} \times \text{Nat}) \text{ outl } \llbracket \text{CoinBagOD} \rrbracket$

$\text{GreedyBagOD} = \text{algOD } \llbracket \text{CoinBagOD} \rrbracket R$

$\text{GreedyBag} : \text{Coin} \rightarrow \text{Nat} \rightarrow \text{Set}$

$\text{GreedyBag } c \ n = \mu \llbracket \text{GreedyBagOD} \rrbracket (c, n)$

whose two constructors can be given the following types:

$\text{gnil} : \{c : \text{Coin}\} \{n : \text{Nat}\} \rightarrow$   
 $\text{total-value-alg } ('nil, \blacksquare) \equiv n \rightarrow$   
 $((ns : \mathbb{F} \llbracket \text{CoinBagOD} \rrbracket (\text{const Nat})) \rightarrow$   
 $\text{total-value-alg } ns \equiv n \rightarrow Q \ ns ('nil, \blacksquare)) \rightarrow$   
 $\text{GreedyBag } c \ n$

$\text{gcons} : \{c : \text{Coin}\} \{n : \text{Nat}\} (d : \text{Coin}) (d \leq c : d \leq_c c) \rightarrow$   
 $\{n' : \text{Nat}\} \rightarrow \text{total-value-alg } ('cons, d, d \leq c, n') \equiv n \rightarrow$   
 $((ns : \mathbb{F} \llbracket \text{CoinBagOD} \rrbracket (\text{const Nat})) \rightarrow$   
 $\text{total-value-alg } ns \equiv n \rightarrow Q \ ns ('cons, d, d \leq c, n')) \rightarrow$   
 $\text{GreedyBag } d \ n' \rightarrow \text{GreedyBag } c \ n$

and implement the greedy algorithm by

$greedy : (c : \text{Coin}) (n : \text{Nat}) \rightarrow \text{GreedyBag } c \ n$

Let us first simplify the two constructors of `GreedyBag`. Each of the two constructors has two additional proof obligations coming from the algebra  $R$ :

- For `gnil`,
  - the first obligation  $total\text{-}value\text{-}alg ('nil, \blacksquare) \equiv n$  reduces to  $0 \equiv n$ , so we may discharge the obligation by specialising  $n$  to 0;
  - for the second obligation,  $ns$  is necessarily  $('nil, \blacksquare)$  if  $total\text{-}value\text{-}alg ns \equiv 0$ , and indeed  $Q$  maps  $('nil, \blacksquare)$  to  $('nil, \blacksquare)$ , so the second obligation can be discharged as well.

We thus obtain a simplified version of `gnil`:

$gnil' : \{c : \text{Coin}\} \rightarrow \text{GreedyBag } c \ 0$

- For `gcons`,
  - the first obligation reduces to  $value \ d + n' \equiv n$ , so we may just specialise  $n$  to  $value \ d + n'$  and discharge the obligation;
  - for the second obligation, any  $ns$  satisfying  $total\text{-}value\text{-}alg ns \equiv value \ d + n'$  must be of the form  $('cons, e, e \leq c, m', \blacksquare)$  for some  $e : \text{Coin}$ ,  $e \leq c : e \leq_C c$ , and  $m' : \text{Nat}$  since the right-hand side  $value \ d + n'$  of the equality is non-zero, and  $Q$  maps  $ns$  to  $('cons, d, d \leq c, n', \blacksquare)$  if  $e \leq_C d$ , so  $d$  should be the largest “usable” coin if this obligation is to be discharged. We say that  $d : \text{Coin}$  is **usable** with respect to some  $c : \text{Coin}$  and  $n : \text{Nat}$  if  $d$  is bounded above by  $c$  and can be part of a solution to the problem for  $n$  pence:

$UsableCoin : \text{Nat} \rightarrow \text{Coin} \rightarrow \text{Coin} \rightarrow \text{Set}$

$UsableCoin \ n \ c \ d = (d \leq_C c) \times (\Sigma [n' : \text{Nat}] \ value \ d + n' \equiv n)$

The obligation can then be rewritten as

$(e : \text{Coin}) \rightarrow UsableCoin \ (value \ d + n') \ c \ e \rightarrow e \leq_C d$

which requires that  $d$  is the largest usable coin with respect to  $c$  and  $value \ d + n'$ . This obligation is the only one that cannot be trivially discharged, since it requires computation of the largest usable coin.

We thus specialise `gcons` to

$gcons' : \{c : \text{Coin}\} (d : \text{Coin}) \rightarrow d \leq_C c \rightarrow$   
 $\{n' : \text{Nat}\} \rightarrow$

$$\begin{aligned} & ((e : \text{Coin}) \rightarrow \text{UsableCoin } (\text{value } d + n') \ c \ e \rightarrow e \leq_c d) \rightarrow \\ & \text{GreedyBag } d \ n' \rightarrow \text{GreedyBag } c \ (\text{value } d + n') \end{aligned}$$

Because of  $\text{gcons}'$ , we are directed to implement a function *maximum-coin* that computes the largest usable coin with respect to any  $c : \text{Coin}$  and non-zero  $n : \text{Nat}$ :

*maximum-coin* :

$$\begin{aligned} & (c : \text{Coin}) (n : \text{Nat}) \rightarrow n > 0 \rightarrow \\ & \Sigma[d : \text{Coin}] \ \text{UsableCoin } n \ c \ d \times ((e : \text{Coin}) \rightarrow \text{UsableCoin } n \ c \ e \rightarrow e \leq_c d) \end{aligned}$$

This takes some theorem proving but is overall a typical AGDA exercise in dealing with natural numbers and ordering. Finally, the greedy algorithm is implemented as the following internalist program, which repeatedly uses *maximum-coin* to find the largest usable coin and unfolds a *GreedyBag*:

$$\begin{aligned} & \text{greedy} : (c : \text{Coin}) (n : \text{Nat}) \rightarrow \text{GreedyBag } c \ n \\ & \text{greedy } c \ n = \text{<-rec } P \ f \ n \ c \\ & \textbf{where} \\ & \quad P : \text{Nat} \rightarrow \text{Set} \\ & \quad P \ n = (c : \text{Coin}) \rightarrow \text{GreedyBag } c \ n \\ & \quad f : (n : \text{Nat}) \rightarrow ((n' : \text{Nat}) \rightarrow n' < n \rightarrow P \ n') \rightarrow P \ n \\ & \quad f \ n \quad \text{rec } c \ \textbf{with} \ \text{compare-with-zero } n \\ & \quad f \ .0 \quad \text{rec } c \mid \text{is-zero} = \text{gnil}' \\ & \quad f \ n \quad \text{rec } c \mid \text{above-zero } n > z \ \textbf{with} \ \text{maximum-coin } c \ n \ n > z \\ & \quad f \ .(\text{value } d + n') \ \text{rec } c \mid \text{above-zero } n > z \mid d, (d \leq_c n', \text{refl}), \text{guc} = \\ & \quad \text{gcons}' \ d \ d \leq_c \text{guc} \ (\text{rec } n' \ \{ \}_8 \ d) \end{aligned}$$

In *greedy*, the combinator

$$\begin{aligned} & \text{<-rec} : (P : \text{Nat} \rightarrow \text{Set}) \rightarrow \\ & \quad ((n : \text{Nat}) \rightarrow ((n' : \text{Nat}) \rightarrow n' < n \rightarrow P \ n') \rightarrow P \ n) \rightarrow \\ & \quad (n : \text{Nat}) \rightarrow P \ n \end{aligned}$$

is for well-founded recursion on  $_{<}$ , and the function

$$\text{compare-with-zero} : (n : \text{Nat}) \rightarrow \text{ZeroView } n$$

is a covering function for the view



```

data ZeroView : Nat → Set where
  is-zero      : ZeroView 0
  above-zero   : {n : Nat} → n > 0 → ZeroView n

```

At Goal 8, AGDA deduces that  $n$  is *value*  $d + n'$  and demands that we prove  $n' < \text{value } d + n'$  in order to make the recursive call, which is easily discharged since *value*  $d > 0$ .

## 5.4 Discussion

Section 5.1 heavily borrows techniques from the AoPA (Algebra of Programming in AGDA) library [Mu et al., 2009] while making generalisations and adaptations: AoPA deals with non-dependently typed programs only, whereas to work with indexed datatypes we need to move to indexed families of relations; to work with the ornamental universe, we parametrise the relational fold with a description, making it fully datatype-generic, whereas AoPA has only specialised versions for lists and binary trees; we define  $\text{min\_} \cdot \Lambda\_$  as a single operator (which happens to be the “shrinking” operator proposed by Mu and Oliveira [2012]) to avoid the struggle with predicativity that AoPA had. All of the above are not fundamental differences between the work presented in this chapter and AoPA, though — the two differ essentially in methodology: in AoPA, dependent types merely provide a foundation on which relational program derivations can be expressed formally and checked by machine, but the programs remain non-dependently typed throughout the formalisation; whereas in this chapter, relational programming is a tool for obtaining nontrivial inductive families that effectively guide program development as advertised as the strength of internalist programming. In short, the focus of AoPA is on traditional relational program derivation (expressed in a dependently typed language), whereas our emphasis is on internalist programming (aided by relational programming).

Algebraic ornamentation was originally proposed by McBride [2011], which deals with functions only. Generalising algebraic ornamentation to a relational setting allows us to write more specifications (like the one given by the Stream-

ing Theorem, which involves converses) and employ more powerful theorems (like the Greedy Theorem, which involves minimisation). We will show in ?? that this generalisation is in fact a “maximal” one: any datatype obtained via ornamentation can be obtained via relational algebraic ornamentation up to isomorphism. Atkey et al. [2012] also investigated algebraic ornamentation via a fibrational, syntax-free perspective. While their “partial refinement” (which generalises functional algebraic ornamentation) is subsumed by relational algebraic ornamentation (since relational algebras allow partiality), they were able to go beyond inductive families to indexed inductive-recursive datatypes, which are out of the scope of this dissertation.

Let us contemplate the interplay between internalist programming and relational programming, especially the one in Section 5.3.3. As mentioned in ??, internalist programs can encode more semantic information, including correctness proofs; we can thus write programs that directly explain their own meaning. The internalist program *greedy* is such an example, whose structure carries an implicit inductive proof; the program constructs not merely a list of coins, but a bag of coins chosen according to a particular local optimisation strategy (i.e.,  $\min Q \cdot \Lambda (\text{fun } \textit{total-value-alg}^\circ)$ ). Internalist programming alone has limited power, however, because internalist programs should share structure with their correctness proofs, but we cannot expect to have such coincidences all the time. In particular, there is no hope of integrating a correctness proof into a program when the structure of the proof is more complicated than that of the program. For example, it is hard to imagine how to integrate a correctness proof for the full specification of the minimum coin change problem into a program for the greedy algorithm. In essence, we have two kinds of proofs to deal with: the first kind follow program structure and can be embedded in internalist programs, and the second kind are general proofs of full specifications, which do not necessarily follow program structure and thus fall outside the scope of internalism. To exploit the power of internalism as much as possible, we need ways to reduce the second kind of proof obligation to the first kind — note that such reduction involves not only constructing proof transformations but also determining what internalist proofs are sufficient for establishing proofs of full specifications. It turns out that relational program derivation is

precisely a way in which to construct such proof transformations systematically from specifications. In relational program derivation, we identify important forms of relational programs (i.e., relational composition, recursion schemes, and various other combinators), and formulate algebraic laws and theorems in terms of these forms. By applying the laws and theorems, we massage a relational specification into a known form which corresponds to a proof obligation that can be expressed in an internalist type, enabling transition to internalist programming. For example, we now know that a relational fold can be turned into an inductive family for internalist programming by algebraic ornamentation. Thus, given a relational specification, we might seek to massage it into a relational fold when that possibility is pointed out by known laws and theorems (e.g., the Greedy Theorem). To sum up, we get a hybrid methodology that leads us from relational specifications towards internalist types for type-directed programming, providing hints in the form of relational algebraic laws and theorems, and this is made possible by going through the synthetic direction of the interconnection between internalism and externalism, synthesising internalist types from relational expressions via algebraic ornamentation.

# Bibliography

- Thorsten ALTENKIRCH, James CHAPMAN, and Tarmo UUSTALU [2010]. Monads need not be endofunctors. In *Foundations of Software Science and Computational Structures*, volume 6014 of *Lecture Notes in Computer Science*, pages 297–311. Springer-Verlag. doi: 10.1007/978-3-642-12032-9\_21. ↗ page 3
- Robert ATKEY, Patricia JOHANN, and Neil GHANI [2012]. Refining inductive types. *Logical Methods in Computer Science*, 8(2:09). doi: 10.2168/LMCS-8(2:9)2012. ↗ page 34
- John BACKUS [1978]. Can programming be liberated from the von Neumann style? A functional style and its algebra of programs. *Communications of the ACM*, 21(8):613–641. doi: 10.1145/359576.359579. ↗ page 2
- Richard BIRD [1996]. Functional algorithm design. *Science of Computer Programming*, 26(1–3):15–31. doi: 10.1016/0167-6423(95)00033-X. ↗ page 2
- Richard BIRD [2010]. *Pearls of Functional Algorithm Design*. Cambridge University Press. ↗ page 2
- Richard BIRD and Oege DE MOOR [1997]. *Algebra of Programming*. Prentice-Hall. ↗ pages 1, 6, 10, 23, and 24
- Richard BIRD and Jeremy GIBBONS [2003]. Arithmetic coding with folds and unfolds. In *Advanced Functional Programming*, volume 2638 of *Lecture Notes in Computer Science*, pages 1–26. Springer-Verlag. doi: 10.1007/978-3-540-44833-4\_1. ↗ pages 10, 15, and 20

- Pierre-Évariste DAGAND and Conor McBRIDE [2012]. Transporting functions across ornaments. In *International Conference on Functional Programming*, ICFP'12, pages 103–114. ACM. doi: 10.1145/2364527.2364544.
- Jeremy GIBBONS [2007]. Metamorphisms: Streaming representation-changers. *Science of Computer Programming*, 65(2):108–139. doi: 10.1016/j.scico.2006.01.006. ↱ pages 15 and 19
- Conor McBRIDE [2011]. Ornamental algebras, algebraic ornaments. To appear in *Journal of Functional Programming*. ↱ pages 9 and 33
- Conor McBRIDE and Ross PATERSON [2008]. Applicative programming with effects. *Journal of Functional Programming*, 18(1):1–13. doi: 10.1017/S0956796807006326. ↱ page 3
- Erik MEIJER, Maarten FOKKINGA, and Ross PATERSON [1991]. Functional programming with bananas, lenses, envelopes and barbed wire. In *Functional Programming Languages and Computer Architecture*, number 523 in Lecture Notes in Computer Science, pages 124–144. Springer-Verlag. doi: 10.1007/3540543961\_7. ↱ page 6
- Eugenio MOGGI [1991]. Notions of computation and monads. *Information and Computation*, 93(1):55–92. doi: 10.1016/0890-5401(91)90052-4. ↱ page 3
- Shin-Cheng MU, Hsiang-Shang Ko, and Patrik JANSSEN [2009]. Algebra of Programming in Agda: Dependent types for relational program derivation. *Journal of Functional Programming*, 19(5):545–579. doi: 10.1017/S0956796809007345. ↱ pages 14 and 33
- Shin-Cheng MU and José Nuno OLIVEIRA [2012]. Programming from Galois connections. *Journal of Logic and Algebraic Programming*, 81(6):680–704. ↱ page 33
- Keisuke NAKANO [2013]. Metamorphism in jigsaw. *Journal of Functional Programming*, 23(2):161–173. doi: 10.1017/S0956796812000391. ↱ page 15
- Bengt NORDSTRÖM [1988]. Terminating general recursion. *BIT Numerical Mathematics*, 28(3):605–619. doi: 10.1007/BF01941137. ↱ page 20

---

Philip WADLER [1992]. The essence of functional programming. In *Principles of Programming Languages*, POPL'92, pages 1–14. ACM. doi: 10.1145/143165.143169. ↗ page 3

## Todo list